MODEIN2 AND COLBY:

COMPUTER CODES FOR SEDIMENT

TRANSPORT COMPUTATIONS

by

V. Miguel Ponce
Jesus Lopez-Garcia
Daryl B. Simons

Colorado State University
Engineering Research Center
Fort Collins, Colorado 80523

November, 1976

CER76-77VMP-JL-DBS19

TABLE OF CONTENTS

I. MODEIN2:

The Modified Einstein Procedure

1. INTRODUCTION

Program MODEIN2 computes the total sediment load and its size distribution in sandbed channels. The procedure used is the Modified Einstein Procedure (MEP) developed by the U.S. Geological Survey [1] and the U.S. Bureau of Reclamation [2,3]. Essentially, the MEP is based on the direct measurement of hydraulic quantities, bed-material size and the suspended load (except within a small distance near the bed) in an alluvial channel. The procedure extrapolates the sediment discharge in the unmeasured zone, thus calculating the total sediment load. The MEP has the same phenomenological structure as Einstein's Bed-Load function [4] with some modifications in the empirical components.

The sediment load computation by the MEP is more accurate than by other computational methods, mainly because the MEP is based on the direct measurement of the hydraulic and sediment transport quantities. This is especially true in sandbed channels where a large proportion of the total sediment load is transported in the sampled zone and is actually measured. The MEP is only applicable where the basic hydraulic and sedimentation parameters have been measured in the field.

2. MAIN FEATURES OF MODEIN2

Program MODEIN2 basically follows the computational procedure outlined in reference (2). However, in order to make the program more reliable, two additional features have been included:

1) The calculation of the Rouse number for fractions other than the reference size is based on the correction suggested on reference (3).

2) The integral functions that are used in the procedure are evaluated by using the algorithm developed by Li (5). The method consists of expanding the integral functions in the form of power series.

With this approach, the computer time is considerably reduced, and the desired degree of accuracy can be selected by the user to satisfy the needs of a particular problem. In the analysis of several test runs with different values of the convergence parameter CONV, a value of CONV=0.01 has generally been found to satisfy both accuracy and computer time requirements.

3. INPUT-OUTPUT DESCRIPTION

MODEIN2 can be set up to read and analyze as many runs as needed. For each series of runs analyzed at one time, the program provides an option to use either the 1:2 ratio sieve sizes in reference [2] or any other series specified by the user.

The output can be limited to the sedimentation quantities related to total load, or extended to print additional hydraulic parameters and intermediate computational values.

Details of input-output controls follow:

A) NUMBER OF SETS CARD. This is the first card in the input record and contains the value of NDATA in format I5. NDATA is the number of sets to be analyzed at one time. Each set of input data consists of a group of variables related to one observation, as detailed below. It should be emphasized that one observation may relate to the computation of the sediment load in the whole of the cross section or the load in a subsection or a vertical, as the case may be.

This first card is to be followed by the individual sets of input data, each consisting of the following.

B-1) GENERAL DATA CARDS. Two cards should be used for the input of the general data (13 variables in format 8F10.0).

The following names are used for the variables:

VARIABLES	FORTRAN NAME	UNITS
Water Discharge	DISCH	cu ft per sec
Average Velocity	UAVE	ft per sec
Hydraulic Depth	DEPTH	ft
Water Surface Width	W	ft
Area of Cross-Section	AREA	sq ft
Temperature	TEMP	°F
Kinematic Viscosity	XNU	sq ft per sec
65 Percent Finer Diameter for Bed Material	D65	ft
35 Percent Finer Diameter for Bed Material	D35	ft
Average Concentration	CONC	ppm
Sampled Suspended Load	QSM	tons per day
Portion of Depth Not Sampled	DN	ft
Average Depth of Sampling	DS	ft

B-2) INPUT-OUTPUT OPTIONS CARD. The values of JIN and JOUT should be punched in format 2I1 according to the following options:

JIN: Selects the number and range in the size fractions that will be analyzed. ND is the number of fractions. The following options can be used:

JIN=1 The size fractions utilized in reference (2) will be used. In this case, the first two size fractions will be used and the third one deleted, hence resulting in ND=10

JIN=2 The size fractions mentioned in reference (2) will also be used, but the first two size fractions will be deleted and the third one used instead, resulting in ND=9.

JIN=3 The user has the option of specifying the number and range of the size fractions to be computed, up to nine fractions. If this option is chosen, then ND should be read immediately after the input-output card, in format I1. Also, changes in the data arrays should be made, as will be detailed below (see B-3).

JOUT: Selects the type of output desired. The following options can be used for JOUT:

JOUT=1 In this case, the output will consist of the general data, the check on convergence of Z prime and the final results in 20 columns, as follows:

1) Geometric mean diameter, in ft
2) PSI
3) PHI shear
4) Percentage of bed material in size fraction
5) Bed load transport, in tons/day
6) Percentage of suspended load in size fraction
7) Sampled transport in size fraction
8) Multipliers
9) A prime values
10) A double prime values
11) Geometric mean diameter, in ft
12) J-one prime
13) J-two prime
14) J-one double prime
15) J-two double prime
16) Product of J's
17) I-one double prime
18) I-two double prime
19) Product of I's
20) Computed load, in tons/day

JOUT=2 Only columns 1, 4, 5, 6 ana 20 will be printed and the rest will be omitted. Additionally, the lower and upper limits of the size fraction range, in mm, DRL(J) and DRU(J), will be printed on the left side of the five previously mentioned columns.

B-3) DATA ARRAYS CARDS. The number of cards and the input depends on the value of JIN.

JIN=1 Ten cards are required in this case, each containing both the values of the fraction of bed material FB(J) and the fraction of suspended load FS(J) punched in format 2F10.0, for each particular size range.

JIN=2 The input consists of nine cards with the same information as in JIN=1 punched in format 2F10.0.

JIN=3 In addition to the percentages FB(J) and FS(J), the range of the computational size fractions should be specified. Hence, DRL(ND), DRU(ND), FB(ND) and FS(ND) should be punched in format 4F10.0, being DRL(J) and DRU(J) the lower and upper limits of each particular size fraction range in mm respectively. Note that size fractions should be punched in the order of increasing size.

A sequence of three runs is illustrated for the following job setup. Different integer selections for JIN and JOUT have been used for illustration. The corresponding output follows the data card assembly example.

4. FORTRAN NAMES FOR INPUT AND OUTPUT VARIABLES

INPUT

Water discharge	DISCH
Average velocity	UAVE
Hydraulic depth	DEPTH
Water surface width	W
Area	AREA
Temperature	TEMP
Kinematic viscosity	XNU
65 percent finer diameter for bed-material	D65
35 percent finer diameter for bed-material	D35
Average concentration	CONC
Sampled suspended load	QSM
Portion of depth not sampled	DN
Average depth of sampling	DS

OUTPUT

Geometric mean diameter, in ft	D(J)
PSI	PSI(J)
PHI shear	PHISH(J)
Percentage of bed-material in size fraction	FB(J)
Bed-load transport, in tons/day	XIBQB(J)
Percentage of suspended load in size fraction	FS(J)
Sampled transport in size fraction	QSP(J)
Multipliers	XMULT(J)
Z prime values	ZP(J)
A double prime values	APP(J)
Geometric mean diameter, in ft	D(J)
J-one prime	COL16(J)
J-two prime	COL17(J)
J-one double prime	COL18(J)
J-two double prime	COL19(J)
Product of J's	COL20(J)
I-one double prime	COL21(J)
I-two double prime	COL22(J)
Product of I's	COL23(J)
Computed load, in tons/day	FQL(J)
Trial Z	ZTRY
Real Q_s'	RQSP
Computed Q_s'	CRQSP
Difference of real and computed Q_s'	DCRQ
Settling velocity	VS(J)
Total bed load	TBL
Total suspended bed material load	TSL
Total bed material load	TQL

5. EXAMPLES



Setup of Data Cards for MODEIN2

COMPUTATION OF TOTAL SEDIMENT LOAD BY THE MODIFIED EINSTEIN PROCEDURE

DATA INPUT

SET 1

WATER DISCHARGE	230.00 C.F.S.
AVERAGE VELOCITY	2.08 FT./SEC.
HYDRAULIC DEPTH	.98 FT.
WATER SURFACE WIDTH	113.00 FT.
AREA	111.00 SQ.FT.
TEMPERATURE	64.00 DEG.FAHREN.
KINEMATIC VISCOSITY	.0000114 SQ.FT./SEC.
D65	.001050 FT.
D35	.000750 FT.
AVERAGE CONCENTRATION	262.00 PPM.
SAMPLED SUSPENDED LOAD	163.0000 TONS/DAY
PORTION OF DEPTH NOT SAMPLED	.30 FT.
AVERAGE DEPTH AT SAMPLING	1.22 FT.

CONVERGENCE OF SUBROUTINE ZPCOM IS CHECKED BY PRINTING OUT VALUES INVOLVED

ITER.	ZTRY	RQSP	CRQSP	DCRQ
1	.80537	6.79554	4.99502	-1.80052
2	.75340	6.79554	7.08348	.28794
3	.75964	6.79554	6.79184	-.00370

ITER.	ZTRY	RQSP	CRQSP	DCRQ
1	1.19938	.43475	.43227	-.00248

ITER.	ZTRY	RQSP	CRQSP	DCRQ
1	1.26422	.27654	.35925	.08271
2	1.31135	.27654	.28385	.00731
3	1.31664	.27654	.27645	-.00009

ARRAYS ZP AND VS BEFORE LEAST SQUARE FIT

J	ZP(J)	VS(J)
3	.759643	.067624
4	1.199380	.152040
5	1.316644	.258550

ARRAYS ZP AND VS AFTER LEAST SQUARE FIT

J	ZP(J)	VS(J)
1	.084548	.000348
2	.476953	.020833
3	.784643	.067624
4	1.105187	.152040
5	1.383333	.258550
6	1.647201	.390728
7	1.926197	.565719
8	2.238023	.806727
9	2.594427	1.144244

J	D(J)	PSI(J)	PHISH(J)	FB(J)	XIBQB(J)	FS(J)	QSP(J)	XMULT(J)	ZP(J)	APP(J)
1	.000037	5.489	.51128	0.000	0.000	.220	28.292	0.000	.085	.000075
2	.000290	5.489	.51128	0.000	0.000	.250	32.150	0.000	.477	.000592
3	.000580	5.489	.51128	.380	7.948	.420	54.012	0.000	.785	.001184
4	.001160	5.489	.51128	.500	29.580	.100	12.860	0.000	1.105	.002367
5	.002320	6.791	.28418	.050	4.650	.010	1.286	0.000	1.383	.004734
6	.004640	13.582	.02402	.010	.222	0.000	0.000	0.000	1.647	.009469
7	.009280	27.163	.00010	.010	.003	0.000	0.000	0.000	1.926	.018938
8	.018559	54.327	.00000	0.000	0.000	0.000	0.000	0.000	2.238	.037876
9	.037118	108.654	.00000	0.000	0.000	0.000	0.000	0.000	2.594	.075752

J	D(J)	COL16(J)	COL17(J)	COL18(J)	COL19(J)	COL20(J)	COL21(J)	COL22(J)	COL23(J)	COMP.LOAD
1	.000037	.729	-.416	1.028	-1.162	1.332	0.000	0.000	0.000	37.6784
2	.000290	.638	-.458	1.483	-3.057	2.011	0.000	0.000	0.000	64.6693
3	.000580	0.000	0.000	0.000	0.000	0.000	2.641	-7.964	21.234	168.7723
4	.001160	0.000	0.000	0.000	0.000	0.000	.831	-3.085	6.792	200.9057
5	.002320	0.000	0.000	0.000	0.000	0.000	.439	-1.664	4.025	18.7158
6	.004640	0.000	0.000	0.000	0.000	0.000	.286	-1.015	3.043	.6767
7	.009280	0.000	0.000	0.000	0.000	0.000	.203	-.638	2.528	.0065
8	.018559	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.0000
9	.037118	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.0000

TOTAL BED LOAD 42.4035 TONS/DAY
TOTAL SUSPENDED LOAD 449.0211 TONS/DAY
TOTAL LOAD 491.4246 TONS/DAY

COMPUTATION OF TOTAL SEDIMENT LOAD BY THE MODIFIED EINSTEIN PROCEDURE

DATA INPUT

SET 2		
WATER DISCHARGE	16300.00	C.F.S.
AVERAGE VELOCITY	4.11	FT./SEC.
HYDRAULIC DEPTH	10.20	FT.
WATER SURFACE WIDTH	389.00	FT.
AREA	3966.00	SQ.FT.
TEMPERATURE	62.00	DEG.FAHREN.
KINEMATIC VISCOSITY	.0000109	SQ.FT./SEC.
D65	.000673	FT.
D35	.000557	FT.
AVERAGE CONCENTRATION	1160.00	PPM.
SAMPLED SUSPENDED LOAD	51051.6000	TONS/DAY
PORTION OF DEPTH NOT SAMPLED	.50	FT.
AVERAGE DEPTH AT SAMPLING	9.70	FT.

CONVERGENCE OF SUBROUTINE ZPCOM IS CHECKED BY PRINTING OUT VALUES INVOLVED

ITER.	ZTRY	RQSP	CRQSP	DCRQ

DRL(J)	DRU(J)	D(J)	FB(J)	XIBQB(J)	FS(J)	FQL(J)
.002000	.062500	.000037	0.000000	0.000000	.800000	40986.697
.062500	.125000	.000290	.040000	7.651876	.100000	9178.504
.125000	.250000	.000580	.820000	443.676883	.100000	11582.569
.250000	.500000	.001160	.140000	214.252539	0.000000	746.373

TOTAL BED LOAD	665.5813	TONS/DAY
TOTAL SUSPENDED LOAD	61828.5613	TONS/DAY
TOTAL LOAD	62494.1426	TONS/DAY

APPENDIX A: REFERENCES

1. Colby, B. R. and Hembree, C. H., "Computations of Total Sediment Discharge, Niobrara River near Cody, Nebraska," U.S. Geological Survey Water Supply Paper 1357, 1955.

2. U.S. Bureau of Reclamation Publication, "Step Method for Computing Total Sediment Load by the Modified Einstein Procedure," July, 1955 (Revised).

3. U.S. Bureau of Reclamation Publication, "Computation of Z for Use in the Modified Einstein Procedure," June, 1966.

4. Einstein, H. A., "The Bed-Load Function for Sediment Transportation in Open Channel Flows," Technical Bulletin 1026, September, 1950, U.S. Department of Agriculture, Soil Conservation Service.

5. Li, R. M., "Mathematical Modeling of Response from Small Watersheds," PhD. Diss. Colorado State University, Fort Collins, Colorado, August, 1974, Appendix A pp. 156-169.

APPENDIX B:

LISTING

```
      PROGRAM MODEIN2 (INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C
C
C     DEVELOPED         COLORADO STATE UNIVERSITY ENGINEERING RESEARCH
C                       CENTER, FORT COLLINS, COLORADO, 80523.
C     PURPOSE           COMPUTATION OF TOTAL SEDIMENT DISCHARGE BY
C                       THE MODIFIED EINSTEIN PROCEDURE.
C     REFERENCES        U.S. BUREAU OF RECLAMATION PUBLICATION
C                       STEP METHOD FOR COMPUTING TOTAL SEDIMENT LOAD
C                       BY THE MODIFIED EINSTEIN PROCEDURE, JULY 1955
C                       (REVISED) AND ADDENDUM  COMPUTATION OF Z FOR USE
C                       IN THE MODIFIED EINSTEIN PROCEDURE, JUNE 1966.
C     CORE USAGE        CDC 6400 SCOPE 3.3 SYSTEM DEFAULT VALUE,
C                       43000 OCTAL.
C     COMPILATION TIME  APPROXIMATELY 8 SEC.
C     CENTRAL PROCESSOR
C     TIME FOR ONE
C     SET OF DATA       APPROXIMATELY 1 SEC.
C
C
C     INPUT AND OUTPUT DESCRIPTION
C
C     THE FIRST CARD IN THE INPUT LOGICAL RECORD SHOULD CONTAIN
C     THE VALUE OF NDATA, IN FORMAT I5.  NDATA IS THE NUMBER OF SETS
C     OF INPUT DATA TO BE FED TO THE COMPUTER AT A TIME.  A SET OF INPUT
C     DATA CONSISTS OF A GROUP OF VARIABLES NECESSARY TO SPECIFY
C     A PROBLEM, AS DETAILED BELOW.
C
C     THE FIRST CARD IS TO BE FOLLOWED BY THE NUMBER OF SETS OF INPUT
C     DATA, EACH ONE CONSISTING OF THE FOLLOWING, IN THE ORDER SHOWN
C     (ORDER IS THE SAME AS THAT USED IN REFERENCE 8)
C
C     1) GENERAL DATA, 13 VARIABLES TO BE PUNCHED IN FORMAT (8F10.0)
C        FOLLOWING IS A LIST OF THE VARIABLES, FORTRAN NAME AND UNITS.
C        WATER DISCHARGE                    DISCH        CFS.
C        AVERAGE VELOCITY                   UAVE         FT./SEC.
C        HYDRAULIC DEPTH                    DEPTH        FT.
C        WATER SURFACE WIDTH                W            FT.
C        AREA                               AREA         SQ.FT.
C        TEMPERATURE                        TEMP         DEG.FARENH.
C        KINEMATIC VISCOSITY                XNU          SQ.FT./SEC.
C        65 PERCENT FINER DIAMETER
C             FOR BED-MATERIAL              D65          FT.
C        35 PERCENT FINER DIAMETER
C             FOR BED-MATERIAL              D35          FT.
C        AVERAGE CONCENTRATION              CONC         PPM.
C        SAMPLED SUSPENDED LOAD             QSM          TONS/DAY
C        PORTION OF DEPTH NOT SAMPLED       DN           FT.
C        AVERAGE DEPTH OF SAMPLING          DS           FT.
C
C     2) INTEGER SELECTORS JIN AND JOUT, TO BE PUNCHED IN FORMAT 2I1.
C        JIN SELECTS THE NUMBER AND RANGE IN THE COMPUTATIONAL
C        SIZE FRACTIONS. ND IS THE NUMBER OF SIZE FRACTIONS.
C        IF JIN=1, THE SIZE FRACTIONS IN THE USBR PUBLICATION WILL BE
C        USED. THE FIRST TWO SIZE FRACTIONS WILL BE USED AND THE THIRD
C        DELETED, RESULTING IN ND= 10
C        IF JIN=2, THE SIZE FRACTIONS IN THE USBR PUBLICATION WILL BE
C        USED. IN THIS CASE THE FIRST TWO SIZE FRACTIONS WILL BE DELETED
C        AND THE THIRD USED INSTEAD, RESULTING IN ND=9
C        IF JIN=3, THE USER HAS THE OPTION OF SPECIFYING THE NUMBER AND
C        RANGE OF COMPUTATIONAL SIZE FRACTIONS. IF THIS OPTION IS
C        CHOSEN, ND SHOULD BE READ IN THE CARD IMMEDIATELY FOLLOWING,
C        IN FORMAT I1.
C        JOUT SELECTS THE TYPE OF OUTPUT DESIRED.
C        IF JOUT=1, OUTPUT WILL CONSIST OF THE GENERAL DATA, CHECK ON
C        CONVERGENCE OF Z PRIME, AND THE FINAL RESULTS IN 20 COLUMNS,
C        AS FOLLOWS.
C
C        1) GEOMETRIC MEAN DIAMETER, IN FT.
C        2) PSI
C        3) PHI SHEAR
C        4) PERCENTAGE OF BED MATERIAL IN SIZE FRACTION
C        5) BED LOAD TRANSPORT, TONS/DAY
C        6) PERCENTAGE OF SUSPENDED LOAD IN SIZE FRACTION
C        7) SAMPLED TRANSPORT IN SIZE FRACTION
C        8) MULTIPLIERS
C        9) Z PRIME VALUES
C        10) A DOUBLE PRIME VALUES
C        11) GEOMETRIC MEAN DIAMETER, IN FT
C        12) J ONE PRIME
C        13) J TWO PRIME
C        14) J ONE DOUBLE PRIME
C        15) J TWO DOUBLE PRIME
C        16) PRODUCT OF JS
C        17) I ONE DOUBLE PRIME
C        18) I TWO DOUBLE PRIME
C        19) PRODUCT OF IS
C        20) COMPUTED LOAD, IN TONS/DAY
C        IF JOUT=2 IS SELECTED, MOST OF THE 20 COLUMNS WILL BE OMITTED
C        IN THE PRINTOUT, AND INSTEAD ONLY COLUMNS 1,4,5,6 AND 20 WILL
C        BE PRINTED. ADDITIONALLY, DRL(J) AND DRU(J), LOWER AND UPPER
C        LIMITS OF THE SIZE FRACTION RANGE, IN MM, WILL BE PRINTED TO
C        THE LEFT OF THE 5 COLUMNS PREVIOUSLY MENTIONED.
C
C     3) DATA ARRAYS.
C        IF JIN=1, THE PERCENT OF BED MATERIAL INSIZE FRACTIONS FB(10),
```

```
C        AND PERCENT OF SUSPENDED LOAD IN SIZE FRACTIONS FS(10)
C        SHOULD BE PUNCHED IN FORMAT 2F10.0
C        IF JIN=2, FB(9) AND FS(9) SHOULD BE PUNCHED IN FORMAT 2F10.0
C        IF JIN=3, THE RANGE OF COMPUTATIONAL SIZE FRACTIONS SHOULD BE
C        SPECIFIED IN ADDITION TO THE PERCENTAGES FB AND FS.
C        IF THIS OPTION IS CHOSEN, DRL(ND), DRU(ND), FB(ND) AND FS(ND)
C        SHOULD BE PUNCHED IN FORMAT 4F10.0
C        DRL(J) AND DRU(J) ARE THE LOWER AND UPPER LIMITS OF THE SIZE
C        FRACTION RANGE, IN MM, RESPECTIVELY. NOTE THAT SIZE FRACTIONS
C        SHOULD BE PUNCHED IN ORDER OF INCREASING SIZE.
C
C
      COMMON /ALL/ DISCH,UAVE,DEPTH,W,AREA,TEMP,XNU,D65,D35,CONC,QSM,DN,
     1DS
      COMMON /ALLB/ D(11),VS(11),FB(10),FS(10),XMULT(10),JIN,JOUT,ND,ND1
     1,ND2
      COMMON /ALLC/ QSP(10),XIBQB(10),FQL(10)
      COMMON /ALLD/ P,AP,APP(10),ZP(10)
      COMMON /ALLE/ DRL(11),DRU(11)
      COMMON /CEF/ CJ0(2),CJ1(2),CJ2(2),CJ3(2),CJ(2),C1(2),C2(2),C3(2),C
     14(2)
      DIMENSION COL16(10), COL17(10), COL18(10), COL19(10), COL20(10), C
     1OL21(10), COL22(10), COL23(10), PSI(10), PHISH(10)
      READ (5,310) NDATA
      DO 290 L=1,NDATA
         IDZ=0
         WRITE (6,360)
         WRITE (6,370)
         CALL INPUT1
         CALL INPUT2
         WRITE (6,380)
         WRITE (6,390) L,DISCH,UAVE,DEPTH,W,AREA,TEMP,XNU,D65,D35,CONC,Q
     1   SM,DN,DS
C
C     CALCULATING HYDRAULIC RADIUS*SLOPE RS, PERCENTAGE OF FLOW SAMPLED
C     PFS, AND SEDIMENT DISCHARGE THROUGH THE SAMPLED ZONE QSPT
C
         CALL RSCOM (X,RS)
         CALL PLATE4 (X,PFS,XKS)
         QSPT=QSM*PFS
C
C     CALCULATING PSI(J)
C
         DO 120 J=1,ND
            XPSI=1.65*D35/RS
            YPSI=0.66*D(J)/RS
            XYPSI=XPSI-YPSI
            IF (XYPSI.LT.0) GO TO 110
            PSI(J)=XPSI
            GO TO 120
  110       PSI(J)=YPSI
  120    CONTINUE
C
C     CALCULATING BED LOAD DISCHARGE XIBQB(J) AND PERCENTAGE OF
C     SUSPENDED MATERIAL IN VARIOUS SIZE FRACTIONS QSP(J)
C
         DO 130 J=1,ND
            XX=PSI(J)
            CALL PLATE5 (XX,YY)
            PHISH(J)=YY
            XIBQB(J)=43.2*W*1200.*PHISH(J)/2.*D(J)**1.5*FB(J)
            QSP(J)=FS(J)*QSPT
  130    CONTINUE
C
C     CALCULATING P, APRIME AP, AND A DOUBLE PRIME APP(J)
C
         DXKS=30.2*X*DEPTH/XKS
         P=2.303*ALOG10(DXKS)
         AP=DN/DS
         DO 140 J=1,ND
            APP(J)=2*D(J)/DEPTH
  140    CONTINUE
         CALL SDR (N,K)
         N1=N+1
         NK=N+K
         WRITE (6,300)
C
C     IF K IS GREATER THAN 2, CONTROL BRANCHES TO STATEMENT 160
C     CALCULATING MULTIPLIERS XMULT(J) , AND ZPRIME ZP(J)
C
         IF (K.GT.2) GO TO 160
         CALL MULCOM (K,N1,NK,KK)
         CALL ZPCOM (KK,IDZ)
         IF (IDZ.EQ.1) GO TO 290
         DO 150 J=1,ND
            ZP(J)=ZP(KK)*XMULT(J)
  150    CONTINUE
         GO TO 200
C
C     CALCULATING ZP AND VS ARRAYS TO BE FED TO LEAST SQUARE SUBROUTINE
C     LSZPVS
C
  160    CONTINUE
         DO 170 J=N1,NK
            CALL ZPCOM (J,IDZ)
            IF (IDZ.EQ.1) GO TO 290
```

```
 170      CONTINUE
          IF (JOUT.EQ.2) GO TO 180
          WRITE (6,320)
          WRITE (6,340)
          WRITE (6,350) (J,ZP(J),VS(J),J=N1,NK)
 180      CONTINUE
          CALL LSZPVS (N1,NK,K,VS,ZP,A,B)
          A=EXP(A)
          DO 190 J=1,ND
             XMULT(J)=0.0
             ZP(J)=A*VS(J)**B
 190      CONTINUE
C
C      CALCULATING SEDIMENT LOAD BY USING MODIFIED EINSTEINS INTEGRAL
C      CHARTS 9,10,11 AND 12
C
 200      CONTINUE
          IF (JOUT.EQ.2) GO TO 220
          IF (K.LT.3) GO TO 210
          WRITE (6,330)
 210      CONTINUE
          IF (JOUT.EQ.2) GO TO 220
          WRITE (6,340)
          WRITE (6,350) (J,ZP(J),VS(J),J=1,ND)
 220      CONTINUE
          TQL=0
          TBL=0
          DO 260 I=1,ND
             XM=APP(I)
             ZM=ZP(I)
             IF (FB(I).LT.0.01.AND.FS(I).LT.0.01) GO TO 240
             IF (FB(I).LT.0.01) GO TO 230
             CALL POWER (ZM,XM,COL21(I),COL22(I),DUM1,DUM2,0.01)
             COL23(I)=P*COL21(I)+COL22(I)+1.
             FQL(I)=XIBQB(I)*COL23(I)
             COL16(I)=0.
             COL17(I)=0.
             COL18(I)=0.
             COL19(I)=0.
             COL20(I)=0.
             GO TO 250
 230         CONTINUE
             CALL POWER (ZM,AP,DUM1,DUM2,COL16(I),COL17(I),0.01)
             CALL POWER (ZM,XM,DUM3,DUM4,COL18(I),COL19(I),0.01)
             COL20(I)=(P*COL18(I)+COL19(I))/(P*COL16(I)+COL17(I))
             FQL(I)=QSP(I)*COL20(I)
             COL21(I)=0.
             COL22(I)=0.
             COL23(I)=0.
             GO TO 250
 240         CONTINUE
             FQL(I)=0.0
             COL16(I)=0.
             COL17(I)=0.
             COL18(I)=0.
             COL19(I)=0.
             COL20(I)=0.
             COL21(I)=0.
             COL22(I)=0.
             COL23(I)=0.
 250         CONTINUE
             TQL=TQL+FQL(I)
             TBL=TBL+XIBQB(I)
 260      CONTINUE
          TSL=TQL-TBL
C
C      PRINTING OUTPUT
C
          WRITE (6,400)
          IF (JOUT.EQ.2) GO TO 270
          WRITE (6,410)
          WRITE (6,420) (J,D(J),PSI(J),PHISH(J),FB(J),XIBQB(J),FS(J),QSP(
     1    J),XMULT(J),ZP(J),APP(J),J=1,ND)
          WRITE (6,430)
          WRITE (6,440) (J,D(J),COL16(J),COL17(J),COL18(J),COL19(J),COL20
     1    (J),COL21(J),COL22(J),COL23(J),FQL(J),J=1,ND)
          GO TO 280
 270      CONTINUE
          WRITE (6,450)
          WRITE (6,460) (DRL(J),DRU(J),D(J),FB(J),XIBQB(J),FS(J),FQL(J),J
     1    =1,ND)
 280      CONTINUE
          WRITE (6,470) TBL,TSL,TQL
 290  CONTINUE
C
C      FORMAT STATEMENTS
C
       STOP
C
 300  FORMAT (//,10X,75H CONVERGENCE OF SUBROUTINE ZPCOM IS CHECKED BY P
     1RINTING OUT VALUES INVOLVED,//)
 310  FORMAT (I5)
 320  FORMAT (//,10X,41H ARRAYS ZP AND VS BEFORE LEAST SQUARE FIT,/)
 330  FORMAT (//,10X,40H ARRAYS ZP AND VS AFTER LEAST SQUARE FIT,/)
 340  FORMAT (//,10X,35H           J        ZP(J)          VS(J),/)
 350  FORMAT (10X,I12,2F12.6)
```

```
 360 FORMAT (1H1)                                                       MOD 2890
 370 FORMAT (40X,70HCOMPUTATION OF TOTAL SEDIMENT LOAD BY THE MODIFIED MOD 2900
    1 EINSTEIN PROCEDURE,///)                                           MOD 2910
 380 FORMAT (32X,10HDATA INPUT,//)                                      MOD 2920
 390 FORMAT (10X,4HSET ,I5,/10X,34HWATER DISCHARGE                     ,FMOD 2930
    112.2,13H C.F.S.       ,/10X,34HAVERAGE VELOCITY                    ,FMOD 2940
    212.2,13H FT./SEC.     ,/10X,34HHYDRAULIC DEPTH                     ,FMOD 2950
    312.2,13H FT.          ,/10X,34HWATER SURFACE WIDTH                 ,FMOD 2960
    412.2,13H FT.          ,/10X,34HAREA                                ,FMOD 2970
    512.2,13H SQ.FT.       ,/10X,34HTEMPERATURE                         ,FMOD 2980
    612.2,13H DEG.FAHREN.  ,/10X,34HKINEMATIC VISCOSITY                 ,FMOD 2990
    712.7,13H SQ.FT./SEC.  ,/10X,34HD65                                 ,FMOD 3000
    812.6,13H FT.          ,/10X,34HD35                                 ,FMOD 3010
    912.6,13H FT.          ,/10X,34HAVERAGE CONCENTRATION               ,FMOD 3020
    *12.2,13H PPM.         ,/10X,34HSAMPLED SUSPENDED LOAD              ,FMOD 3030
    *12.4,13H TONS/DAY     ,/10X,34HPORTION OF DEPTH NOT SAMPLED        ,FMOD 3040
    *12.2,13H FT.          ,/10X,34HAVERAGE DEPTH AT SAMPLING           ,FMOD 3050
    *12.2,4H FT.)                                                       MOD 3060
 400 FORMAT (//)                                                        MOD 3070
 410 FORMAT (5X,1HJ,11X,4HD(J),7X,6HPSI(J),5X,8HPHISH(J),7X,5HFB(J),4X,MOD 3080
    18HXIBQB(J),7X,5HFS(J),6X,6HQSP(J),4X,8HXMULT(J),7X,5HZP(J),5X,6HAPMOD 3090
    2P(J)/)                                                             MOD 3100
 420 FORMAT (4X,I2,4X,F12.6,F12.3,F12.5,6F12.3,F12.6)                   MOD 3110
 430 FORMAT (//5X,1HJ,11X,4HD(J),6X,8HCOL16(J),4X,8HCOL17(J),4X,8HCOL18MOD 3120
    1(J),4X,8HCOL19(J),4X,8HCOL20(J),4X,8HCOL21(J),4X,8HCOL22(J),4X,8HCMOD 3130
    2OL23(J),4X,9HCOMP.LOAD/)                                           MOD 3140
 440 FORMAT (4X,I2,4X,F12.6,8F12.3,F12.4)                               MOD 3150
 450 FORMAT (10X,84H       DRL(J)       DRU(J)         D(J)          FB(J)  MOD 3160
    1 XIBQB(J)        FS(J)        FQL(J),/)                            MOD 3170
 460 FORMAT (10X,6F12.6,F12.3)                                          MOD 3180
 470 FORMAT (///,5X,34HTOTAL BED LOAD                         ,F16.4,9H TONSMOD 3190
    1/DAY,/5X,34HTOTAL SUSPENDED LOAD                ,F16.4,9H TONS/DAY,/MOD 3200
    25X,34HTOTAL LOAD                         ,F16.4,9H TONS/DAY)       MOD 3210
C                                                                       MOD 3220
      END                                                               MOD 3230
```

```
      SUBROUTINE INPUT1
C
C     THIS SUBROUTINE READS IN THE BASIC VARIABLES OF THE PROBLEM
C
      COMMON /ALL/ DISCH,UAVE,DEPTH,W,AREA,TEMP,XNU,D65,D35,CONC,QSM,DN,
     1DS
      READ (5,110) DISCH,UAVE,DEPTH,W,AREA,TEMP,XNU,D65,D35,CONC,QSM,DN,
     1DS
      RETURN
C
  110 FORMAT (8F10.0)
C
      END
```

```
      SUBROUTINE INPUT2
C
C     THIS SUBROUTINE READS IN ADDITIONAL INPUT AND FINDS THE VALUE OF
C     ND, THE NUMBER OF SIZE FRACTIONS TO BE USED IN THE COMPUTATION
C
      COMMON /ALL/ DISCH,UAVE,DEPTH,W,AREA,TEMP,XNU,D65,D35,CONC,QSM,DN,
     1DS
      COMMON /ALLB/ D(11),VS(11),FB(10),FS(10),XMULT(10),JIN,JOUT,ND,ND1
     1,ND2
      COMMON /ALLE/ DRL(11),DRU(11)
      DRL(1)=.002$DRL(2)=.0156$DRL(3)=.002 $DRL(4)=.0625$DRL(5)=.125
      DRL(6)=.25$DRL(7)=.5$DRL(8)=1.$DRL(9)=2.$DRL(10)=4.$DRL(11)=8.
      DRU(1)=.0156$DRU(2)=.0625$DRU(3)=.0625$DRU(4)=.125$DRU(5)=.25
      DRU(6)=.5$DRU(7)=1.$DRU(8)=2.$DRU(9)=4.$DRU(10)=8.$DRU(11)=16.
      READ (5,190) JIN,JOUT
      IF (JIN.EQ.3) GO TO 160
      DO 110 J=1,11
          D(J)=(DRL(J)*DRU(J))**0.5/304.8
          VS(J)=((2./3.*32.17*1.65*D(J)**3.+36.*XNU**2.)**0.5-6.*XNU)/D(J
     1    )
  110 CONTINUE
      ND1=10
      ND2=9
      IF (JIN.EQ.2) GO TO 130
      DO 120 J=3,ND1
          D(J)=D(J+1)
          VS(J)=VS(J+1)
  120 CONTINUE
      ND=ND1
      GO TO 150
  130 DO 140 J=1,ND2
          D(J)=D(J+2)
          VS(J)=VS(J+2)
  140 CONTINUE
      ND=ND2
  150 CONTINUE
      READ (5,200) (FB(J),FS(J),J=1,ND)
      GO TO 180
  160 CONTINUE
      READ (5,210) ND
      READ (5,220) (DRL(J),DRU(J),FB(J),FS(J),J=1,ND)
      DO 170 J=1,ND
          D(J)=(DRU(J)*DRL(J))**0.5/304.8
          VS(J)=((2./3.*32.17*1.65*D(J)**3.+36.*XNU**2.)**0.5-6.*XNU)/D(J
     1    )
  170 CONTINUE
  180 CONTINUE
      RETURN
C
  190 FORMAT (2I1)
  200 FORMAT (2F10.0)
  210 FORMAT (I1)
  220 FORMAT (4F10.0)
C
      END
```

```
      SUBROUTINE RSCOM (X,RS)
C
C     THIS SUBROUTINE COMPUTES THE VALUE OF RS BY ITERATION
C
      COMMON /ALL/ DISCH,UAVE,DEPTH,W,AREA,TEMP,XNU,D65,D35,CONC,QSM,DN,
     1DS
      X=1.6
      TOL=0.001
      XKS=D65
  110 XDKS=12.27*X*DEPTH/XKS
      SRRS=UAVE/(32.65*ALOG10(XDKS))
      USHP=SRRS*5.68
      DEL=11.6*XNU/USHP
      DELKS=XKS/DEL
      CALL PLATE3 (DELKS,X2)
      DELX=X-X2
      IF (ABS(DELX).LT.TOL) GO TO 120
      X=X2
      GO TO 110
  120 CONTINUE
      XDKS=12.27*X*DEPTH/XKS
      SRRS=UAVE/(32.65*ALOG10(XDKS))
      RS=SRRS*SRRS
      RETURN
C
      END
```

```
      SUBROUTINE PLATE4 (X,PFS,XKS)
C
C     THIS SUBROUTINE SUBSTITUTES PLATE FOUR FOR THE ANALYTICAL
C     EXPRESSION OF PFS
C
      COMMON /ALL/ DISCH,UAVE,DEPTH,W,AREA,TEMP,XNU,D65,D35,CONC,QSM,DN,
     1DS
      XKS=D65
      A=30.2*X/XKS
      YDS=DS*ALOG(A*DS)-DS
      YDN=DN*ALOG(A*DN)-DN
      PFS=(YDS-YDN)/YDS
      RETURN
C
      END
```

```
      SUBROUTINE SDR (N,K)
C
C     THIS SUBROUTINE COUNTS THE NUMBER OF SIZE FRACTIONS K FOR WHICH
C     THERE IS BOTH BED AND SUSPENDED DISCHARGE, AND THE NUMBER OF SIZE
C     FRACTIONS N SMALLER THAN FIRST K.
C
      COMMON /ALLB/ D(11),VS(11),FB(10),FS(10),XMULT(10),JIN,JOUT,ND,ND1
     1,ND2
      J=0
      K=0
      N=0
  110 CONTINUE
      IF (FB(J+1).GT.0.00.AND.FS(J+1).GT.0.00) GO TO 130
      IF (K.NE.0) GO TO 120
      N=N+1
  120 J=J+1
      IF (J.EQ.ND) RETURN
      GO TO 110
  130 CONTINUE
      K=K+1
      J=J+1
      IF (J.EQ.ND) RETURN
      GO TO 110
C
      END
```

```
      SUBROUTINE LSZPVS (N1,NK,K,X,Y,A,B)
C
C     THIS SUBROUTINE CALCULATES A LEAST SQUARE FIT FOR ZPRIME ZP(K) AND
C     VS(K)
C
      DIMENSION X(11), Y(10)
      SUMX=0.
      SUMY=0.
      SUMXY=0.
      SUMX2=0.
      DO 110 J=N1,NK
         XL=ALOG(X(J))
         SUMX=SUMX+XL
         YL=ALOG(Y(J))
         SUMY=SUMY+YL
         XY=XL*YL
         SUMXY=SUMXY+XY
         X2=XL*XL
         SUMX2=SUMX2+X2
  110 CONTINUE
      XMEAN=SUMX/K
      YMEAN=SUMY/K
      B=(SUMXY-SUMX*SUMY/K)/(SUMX2-SUMX*SUMX/K)
      A=YMEAN-B*XMEAN
      RETURN
C
      END
```

```
      SUBROUTINE MULCOM (K,N1,NK,KK)
C
C     THIS SUBROUTINE CALCULATES THE MULTIPLIERS XMULT(J)
C
      COMMON /ALLB/ D(11),VS(11),FB(10),FS(10),XMULT(10),JIN,JOUT,ND,ND1
     1,ND2
      DIMENSION SBS(9)
      IF (K.EQ.0) GO TO 160
      IF (K.EQ.2) GO TO 110
      KK=N1
      GO TO 140
  110 CONTINUE
      DO 120 J=N1,NK
         SBS(J)=FB(J)+FS(J)
  120 CONTINUE
      IF (SBS(N1).GT.SBS(NK)) GO TO 130
      KK=NK
      GO TO 140
  130 KK=N1
  140 CONTINUE
      DO 150 J=1,ND
         XMULT(J)=(VS(J)/VS(KK))**0.7
  150 CONTINUE
      GO TO 170
  160 WRITE (6,180)
  170 CONTINUE
      RETURN
C
  180 FORMAT (10X,97HBECAUSE NO SIZE FRACTION CONTAINS BOTH BED AND SUSP
     1ENDED DISCHARGE, THE COMPUTATIONS ARE ABORTED.)
C
      END
```

```
      SUBROUTINE PLATE8 (X,Y)
C
C     THIS SUBROUTINE APPROXIMATES PLATE 8 BY A LINE IN LOG-LOG PAPER
C
      Y=-0.33*ALOG10(X)+1.08
      RETURN
C
      END
```

```
      SUBROUTINE ZPCOM (J,IDZ)
C
C     THIS SUBROUTINE COMPUTES ZPRIME ZP BY ITERATION
C     FIRST, ATRIAL VALUE OF ZP IS CALCULATED, AND THEN, WITH ANOTHER
C     TRIAL, A LINEAR INTERPOLATION IS MADE. CONVERGENCE IS VERY FAST.
C
      COMMON /ALLB/ D(11),VS(11),FB(10),FS(10),XMULT(10),JIN,JOUT,ND,ND1
     1,ND2
      COMMON /ALLC/ QSP(10),XIBQB(10),FQL(10)
      COMMON /ALLD/ P,AP,APP(10),ZP(10)
      XM=APP(J)
      RQSP=QSP(J)/XIBQB(J)
      IF (RQSP.LT.1873.) GO TO 110
      IDZ=1
      WRITE (6,160)
      RETURN
  110 CALL PLATE8 (RQSP,ZTRY)
      STEP=0.01
      WRITE (6,170)
      KOUNT=0
  120 CONTINUE
      KOUNT=KOUNT+1
      IF (KOUNT.GT.10) GO TO 140
      CALL POWER (ZTRY,XM,XI1PP,DUM1,XJ1PP,DUM2,0.01)
      CALL POWER (ZTRY,AP,DUM3,DUM4,XJ1P,XJ2P,0.01)
      CRQSP=XI1PP/XJ1PP*(P*XJ1P+XJ2P)
      DCRQ=CRQSP-RQSP
      IF (JOUT.EQ.2) GO TO 130
      WRITE (6,190) KOUNT,ZTRY,RQSP,CRQSP,DCRQ
  130 CONTINUE
      TOL=0.01*RQSP
      IF (ABS(DCRQ).LT.TOL) GO TO 150
      IF (CRQSP.LT.RQSP) ZTRY1=ZTRY-STEP
      IF (CRQSP.GT.RQSP) ZTRY1=ZTRY+STEP
      CALL POWER (ZTRY1,XM,XI1PP,DUM1,XJ1PP,DUM2,0.01)
      CALL POWER (ZTRY1,AP,DUM3,DUM4,XJ1P,XJ2P,0.01)
      CRQSP1=XI1PP/XJ1PP*(P*XJ1P+XJ2P)
      TEMP=(RQSP-CRQSP)*STEP/(CRQSP1-CRQSP)
      IF (CRQSP.LT.RQSP) ZTRY=ZTRY-TEMP
      IF (CRQSP.GT.RQSP) ZTRY=ZTRY+TEMP
      GO TO 120
  140 CONTINUE
      WRITE (6,180)
  150 CONTINUE
      ZP(J)=ZTRY
      RETURN
C
  160 FORMAT (///,10X,88HRQSP OUT OF PERMISSIBLE RANGE IN THIS SET OF DA
     1TA.CALCULATIONS FOR THIS SET ARE ABORTED.)
  170 FORMAT (//,20X,5HITER.,5X,5HZTRY ,6X,4HRQSP,8X,5HCRQSP,7X,4HDCRQ)
  180 FORMAT (/10X,76HZPCOM DOES NOT CONVERGE WITH 10 ITERATIONS, LAST V
     1ALUE OF ZP(J) WILL BE USED,/)
  190 FORMAT (/10X,I12,4F12.5)
C
      END

      SUBROUTINE PLATE3 (X,Y)
C
C     THIS SUBROUTINE APPROXIMATES PLATE 3 BY A SERIES OF EQUATIONS
C
      IF (X.LE.0.40) GO TO 110
      GO TO 120
  110 Y=1.769*ALOG10(X/0.080)
      GO TO 270
  120 IF (X.GT.0.40.AND.X.LE.0.56) GO TO 130
      GO TO 140
  130 Y=1.495*ALOG10(X/0.059)
      GO TO 270
  140 IF (X.GT.0.56.AND.X.LE.0.76) GO TO 150
      GO TO 160
  150 Y=0.92*ALOG10(X/0.0145)
      GO TO 270
  160 IF (X.GT.0.76.AND.X.LE.0.96) GO TO 170
      GO TO 180
  170 Y=0.292*ALOG10(X/2.9E-06)
      GO TO 270
  180 IF (X.GT.0.96.AND.X.LE.1.35) GO TO 190
      GO TO 200
  190 Y=0.277*ALOG10(632000.0/X)
      GO TO 270
  200 IF (X.GT.1.35.AND.X.LE.3.00) GO TO 210
      GO TO 220
  210 Y=1.115*ALOG10(34.4/X)
      GO TO 270
  220 IF (X.GT.3.00.AND.X.LE.4.00) GO TO 230
      GO TO 240
  230 Y=0.725*ALOG10(128.0/X)
      GO TO 270
  240 IF (X.GT.4.00.AND.X.LE.6.70) GO TO 250
      GO TO 260
  250 Y=0.399*ALOG10(2160.0/X)
      GO TO 270
  260 IF (X.GT.6.70) Y=1.0
  270 RETURN
C
      END
```

```
      SUBROUTINE PLATE5 (X,Y)
C
C     THIS SUBROUTINE APPROXIMATES PLATE 5 BY A SERIES OF EQUATIONS
C
C
      IF (X.LE.0.77) Y=(7.56/X)**1.01
      IF (X.GT.0.77.AND.X.LE.2.12) Y=(5.35/X)**1.19
      IF (X.GT.2.12.AND.X.LE.4.10) Y=(4.10/X)**1.67
      IF (X.GT.4.10.AND.X.LE.6.10) Y=(4.10/X)**2.30
      IF (X.GT.6.10.AND.X.LE.11.0) Y=(4.60/X)**3.23
      IF (X.GT.11.0.AND.X.LE.16.7) Y=(5.66/X)**4.26
      IF (X.GT.16.7.AND.X.LE.22.5) Y=(9.28/X)**7.81
      IF (X.GT.22.5) Y=(13.10/X)**12.66
      RETURN
C
      END
```

```
      SUBROUTINE POWER (Z,A,XI1,XI2,XJ1,XJ2,CONV)
C
C     THIS SUBROUTINE EVALUATES I1 I2 J1 AND J2 INTEGRALS
C     NOTATIONS
C     XI1 = VALUE OF I1 INTEGRAL
C     XI2 = VALUE OF I2 INTEGRAL
C     XJ1 = VALUE OF J1 INTEGRAL
C     XJ2 = VALUE OF J2 INTEGRAL
C     N = ORDER OF APPROXIMATION + 1
C     CONV = CONVERGENCE CRITERION
C
      N=1
      FACT=0.216*A**(Z-1.)/(1.-A)**Z
      XI1=0.
      XI2=0.
      XJ1=0.
      XJ2=0.
      ALG=ALOG(A)
      C=1.
      D=-Z
      E=D+1.
      FN=1.
      AEX=A**E
      GO TO 120
  110 N=N+1
      C=C*D/FN
      D=E
      E=D+1.
      FN=FLOAT(N)
      AEX=A**E
  120 IF (ABS(E).LE.0.001) GO TO 130
      XJ1=XJ1+C*(1.-AEX)/E
      XJ2=XJ2+C*((AEX-1.)/E**2-AEX*ALG/E)
      GO TO 140
  130 XJ1=XJ1-C*ALG
      XJ2=XJ2-0.5*C*ALG**2
  140 IF (N.EQ.1) GO TO 150
      CJ1=ABS(1.-FJ1/XJ1)
      CJ2=ABS(1.-FJ2/XJ2)
      IF (CJ1.LE.CONV.AND.CJ2.LE.CONV) GO TO 160
  150 FJ1=XJ1
      FJ2=XJ2
      GO TO 110
  160 XI1=FACT*XJ1
      XI2=FACT*XJ2
      RETURN
C
      END
```

II. COLBY:

Colby's Bed Material Load Method

1. INTRODUCTION

Program COLBY computes bed material load by Colby's Method [1]. Data input consists of average velocity (ft per sec), hydraulic depth (ft), water surface width (ft), temperature (°F), median bed material size (mm) and fine material concentration (ppm). A remark included as part of the output indicates whether the computations were carried out in a normal fashion, or if one or more variables were out of the value range specified in this method. If velocity, depth or bed material size are out of range, the program fails to give any results. If temperature or fine material concentration are out of range, the program extrapolates and gives a result, albeit of limited value.

2. INPUT-OUTPUT DESCRIPTION

INPUT:

A) NUMBER OF SETS CARD

It is the first card in the input logical record and should contain the value of NDATA, in format I5. NDATA is the number of sets of input data to be fed to the computer at a time. A set of input data consists of a group of variables necessary to specify a problem, as detailed below.

B) INPUT DATA CARDS

The first card in input is followed by the sets of input data, to be punched in format 6F10.0. A set of input data consists of the following variables, relating to a channel cross section.

VARIABLES	FORTRAN NAME	UNITS
1) Average Velocity	V	ft per sec
2) Hydraulic Depth	D	ft
3) Water Surface Width	W	ft
4) Temperature	TF	°F
5) Median Bed Material Size	D50	mm
6) Fine Material Concentration	FML	ppm

OUTPUT:

Output consists of the total bed material transport in tons/day, and a remark on how the computations were carried out. If REMARK=OK, the computations were carried out successfully. If REMARK=OOR, velocity, depth or bed material size is out of range. If REMARK=TOOR, temperature is out of range. If REMARK=FOOR, fine material concentration is out of range.

VARIABLE	RANGE
Average Velocity	1-10 ft per sec
Hydraulic Depth	1-100 ft
Temperature	32-100 °F
Median Bed Material Size	0.1-0.8 mm
Fine Material Concentration	0-200000 ppm

3. FORTRAN NAMES FOR INPUT AND OUTPUT VARIABLES

VARIABLE	FORTRAN NAME
Average Velocity	V
Hydraulic Depth	D
Water Surface Width	W
Temperature	TF
Median Bed Material Size	D50
Fine Material Concentration	FML
Bed Material Transport	GT

4. EXAMPLES



Setup of Data Cards for COLBY

```
COMPUTATION OF TOTAL BED MATERIAL
TRANSPORT BY COLBYS METHOD

SET       1
AVERAGE VELOCITY                        9.92 FT./SEC.
HYDRAULIC DEPTH                         4.14 FT.
WATER SURFACE WIDTH                   234.00 FT.
TEMPERATURE                            70.00 DEG.FAHREN.
MEDIAN BED MATERIAL SIZE                 .32 MM.
FINE MATERIAL CONCENTRATION         10000.00 PPM.

BED MATERIAL TRANSPORT =     76173.08304 TONS/DAY
REMARK = OK

SET       2
AVERAGE VELOCITY                       11.00 FT./SEC.
HYDRAULIC DEPTH                         4.14 FT.
WATER SURFACE WIDTH                   234.00 FT.
TEMPERATURE                            70.00 DEG.FAHREN.
MEDIAN BED MATERIAL SIZE                 .32 MM.
FINE MATERIAL CONCENTRATION         10000.00 PPM.

COMPUTATIONS COULD NOT BE CARRIED OUT
DUE TO DATA OUT OF RANGE
REMARK= OOR

SET       3
AVERAGE VELOCITY                        9.92 FT./SEC.
HYDRAULIC DEPTH                         4.14 FT.
WATER SURFACE WIDTH                   234.00 FT.
TEMPERATURE                           105.00 DEG.FAHREN.
MEDIAN BED MATERIAL SIZE                 .32 MM.
FINE MATERIAL CONCENTRATION         10000.00 PPM.

BED MATERIAL TRANSPORT =     59231.54605 TONS/DAY
REMARK = TOOR

SET       4
AVERAGE VELOCITY                        9.92 FT./SEC.
HYDRAULIC DEPTH                         4.14 FT.
WATER SURFACE WIDTH                   234.00 FT.
TEMPERATURE                            70.00 DEG.FAHREN.
MEDIAN BED MATERIAL SIZE                 .32 MM.
FINE MATERIAL CONCENTRATION        300000.00 PPM.

BED MATERIAL TRANSPORT =    810518.47909 TONS/DAY
REMARK = FOOR
```

APPENDIX A: REFERENCES

1. Colby, B. R., "Discharge of Sands and Mean Velocity Relationships in Sand-bed Streams," U.S. Geological Survey Prof. Paper 462-A, 1964.

APPENDIX B:

LISTING

```
      PROGRAM COLBY (INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C
C
C     DEVELOPED            COLORADO STATE UNIVERSITY ENGINEERING RESEARCH
C                          CENTER, FORT COLLINS,COLORADO 80523
C     PURPOSE              COMPUTATION OF BED MATERIAL LOAD BY COLBYS
C                          METHOD
C     REFERENCE            COLBY,B.R., DISCHARGE OF SANDS AND MEAN VELOCITY
C                          RELATIONSHIPS IN SAND-BED STREAMS, PROFESSIONAL
C                          PAPER 462-A, 1964, U.S. GEOLOGICAL SURVEY.
C     CORE USAGE           CDC 6400 SCOPE 3.3 SYSTEM DEFAULT VALUE,
C                          43000 OCTAL.
C     COMPILATION TIME     APPROXIMATELY 4 SEC.
C     CENTRAL PROCESSOR
C     TIME FOR ONE
C     SET OF DATA          LESS THAN 0.6 SEC.
C
C     INPUT AND OUTPUT DESCRIPTION
C
C     THE FIRST CARD IN THE INPUT LOGICAL RECORD SHOULD CONTAIN THE
C     VALUE OF NDATA, IN FORMAT I5. NDATA IS THE NUMBER OF SETS OF INPUT
C     DATA TO BE FED TO THE COMPUTER AT A TIME. A SET OF INPUT DATA
C     CONSISTS OF A GROUP OF VARIABLES NECESSARY TO SPECIFY A PROBLEM,
C     AS DETAILED BELOW.
C
C     THE FIRST CARD IN INPUT IS FOLLOWED BY THE SETS OF INPUT DATA,
C     TO BE PUNCHED IN FORMAT 6F10.0
C     A SET OF INPUT DATA CONSISTS OF THE FOLLOWING VARIABLES.
C     1) AVERAGE VELOCITY                V                 F.P.S.
C     2) HYDRAULIC DEPTH                 D                 FT.
C     3) WATER SURFACE WIDTH             W                 FT.
C     4) TEMPERATURE                     TF                DEG.FAHREN.
C     5) MEDIAN BED MATERIAL SIZE        D50               MM.
C     6) FINE MATERIAL CONCENTRATION     FML               PPM.
C
C     OUTPUT CONSISTS OF THE TOTAL BED MATERIAL TRANSPORT IN TONS/DAY,
C     AND A REMARK ON HOW THE COMPUTATIONS WERE CARRIED OUT.
C     IF  REMARK= OK, THE COMPUTATIONS WERE CARRIED OUT SUCCESSFULLY.
C     IF  REMARK= OOR, VELOCITY, DEPTH OR BED MATERIAL SIZE IS OUT OF
C     RANGE.
C     IF  REMARK= TOOR, TEMPERATURE IS OUT OR RANGE.
C     IF  REMARK= FOOR, FINE MATERIAL CONCENTRATION IS OUT OF RANGE.
C     VARIABLE                           RANGE
C     AVERAGE VELOCITY                   1-10  F.P.S.
C     HYDRAULIC DEPTH                    1-100  FT.
C     WATER SURFACE WIDTH
C     TEMPERATURE                        32-100  DEG.FAHREN.
C     MEDIAN BED MATERIAL SIZE           0.1-0.8  MM.
C     FINE MATERIAL CONCENTRATION        0-200000 PPM.
C
C
      COMMON /CLBY/ G(4,8,6),F(5,10),T(7,4),P(11),DF(10),CF(5),DP(11),DG
     1(4),VG(8),D50G(6),TEMP(7)
      DIMENSION II(2), JJ(2), KK(2), XX(2), YY(2), ZZ(2), X(2,2), XA(2),
     1 XG(2), XT(2,2), XCT(2), XF(2,2)
      WRITE (6,159)
      READ (5,162) NDATA
      DO 157 L=1,NDATA
          READ (5,163) V,D,W,TF,D50,FML
          WRITE (6,160) L,V,D,W,TF,D50,FML
          DATA (((G(I,J,K),I=1,4),J=1,8),K=1,3)/1.0,0.30,0.06,0.0,3.0,3.3
     1    0,2.50,2.0,5.4,9.00,10.00,20.00,11.0,26.00,50.00,150.00,17.0,49
     2    .00,130.00,500.0,29.0,101.00,400.00,1350.0,44.0,160.00,700.00,2
     3    500.0,60.0,220.0,1000.0,4400.0,.38,0.06,0.00,0.00,1.60,1.20,.65
     4    ,.10,3.70,5.00,4.00,3.00,10.00,18.00,30.00,52.00,17.00,40.00,80
     5    .00,160.00,36.00,95.00,230.00,650.00,60.00,150.00,415.00,1200.0
     6    ,81.00,215.00,620.00,1500.0,0.14,0.0,0.0,0.0,1.00,0.60,.15,0.0,
     7    3.30,3.00,1.7,0.5,11.00,15.00,17.0,14.0,20.00,35.00,49.0,70.0,4
     8    4.00,85.00,150.0,250.0,71.00,145.00,290.0,500.0,100.00,202.00,4
     9    00.0,700.0/
          DATA (((G(I,J,K),I=1,4),J=1,8),K=4,6)/0.0,0.0,0.00,0.00,0.70,0.
     1    30,0.06,0.00,2.90,2.30,1.00,0.05,11.50,13.00,12.00,7.00,22.00,3
     2    1.00,40.00,50.00,47.00,84.00,135.00,210.00,75.00,140.00,240.00,
     3    410.00,106.00,190.00,350.00,630.00,0.00,0.0,0.00,0.0,0.44,0.06,
     4    0.00,0.0,2.80,1.80,0.60,0.0,12.00,12.50,10.00,4.5,24.00,30.00,3
     5    5.00,37.0,52.00,78.00,120.00,190.00,83.00,180.00,215.00,380.0,1
     6    20.00,190.00,305.00,550.00,0.0,0.0,0.0,0.0,0.3,0.0,0.0,0.0,2.9,
     7    1.4,0.3,0.0,14.0,11.0,7.7,3.0,27.0,29.0,30.0,30.0,57.0,75.0,110
     8    .0,170.0,90.0,140.0,200.0,330.0,135.0,190.0,290.0,520.0/
          DATA ((F(I,J),I=2,5),J=1,10)/1.10,1.60,2.60,4.20,1.10,1.65,2.75
     1    ,4.90,1.10,1.70,3.00,5.50,1.12,1.90,3.60,7.00,1.17,2.05,4.30,8.
     2    70,1.20,2.30,5.50,11.20,1.22,2.75,8.00,22.00,1.25,3.00,9.60,29.
     3    00,1.30,3.50,12.00,43.00,1.40,4.90,22.00,120.0/
          DATA (F(1,I),I=1,10)/1.0,1.0,1.0,1.0,1.0,1.0,1.0,1.0,1.0,1.0/
          DATA ((T(I,J),I=1,7),J=1,4)/1.20,1.15,1.10,0.96,0.90,0.85,0.82,
     1    1.35,1.25,1.12,0.92,0.86,0.80,0.75,1.60,1.40,1.20,0.89,0.80,0.7
     2    2,0.66,2.00,1.65,1.30,0.85,0.72,0.63,0.55/
          DATA (DF(I),I=1,10)/0.10,0.20,0.30,0.60,1.00,2.00,6.00,10.00,20
     1    .00,1.E2/,(CF(I),I=1,5)/9.0,1.E4,5.E4,1.E5,1.5E5/
          DATA (P(I),I=1,11)/0.60,0.90,1.0,1.0,0.83,0.60,0.40,0.25,0.15,0
     1    .09,0.05/,(DP(I),I=1,11)/0.10,0.15,0.20,0.30,0.40,0.50,0.60,0.7
     2    0,0.80,0.90,1.00/,(DG(I),I=1,4)/0.10,1.0,10.0,100.0/,(VG(I),I=1
     3    ,8)/1.0,1.5,2.0,3.0,4.0,6.0,8.0,10.0/,(D50G(I),I=1,6)/0.10,0.20
     4    ,0.30,0.40,0.60,0.80/,(TEMP(I),I=1,7)/32.0,40.0,50.0,70.0,80.0,
     5    90.0,100.0/
          REMARK=5HOK
```

```
      IF ((D50.LT.D50G(1)).OR.(D50.GT.D50G(6))) GO TO 101
      GO TO 102
  101 REMARK=5HOOR
      GO TO 155
C
C     LOCATE APPROPRIATE   V,D,D50 GRID
C
  102 CONTINUE
      IF ((D.LT.DG(1)).OR.(D.GT.DG(4))) GO TO 103
      GO TO 104
  103 REMARK=5HOOR
      GO TO 155
  104 IF ((V.LT.VG(1)).OR.(V.GT.VG(8))) GO TO 105
      GO TO 106
  105 REMARK=5HOOR
      GO TO 155
  106 IF (TF.EQ.0.) TF=60.
      IF (TF-32.) 107,108,108
  107 REMARK=5HTOOR
      TF=32.
  108 IF (TF-100) 110,110,109
  109 REMARK=5HTOOR
      TF=100.
  110 CONTINUE
      ID1=0
      ID2=0
      DO 113 I=1,3
         IF ((D.GE.DG(I)).AND.(D.LE.DG(I+1))) GO TO 111
         GO TO 112
  111    ID1=I
         ID2=I+1
         GO TO 114
  112    CONTINUE
  113 CONTINUE
  114 IV1=0
      IV2=0
      DO 117 I=1,7
         IF ((V.GE.VG(I)).AND.(V.LE.VG(I+1))) GO TO 115
         GO TO 116
  115    IV1=I
         IV2=I+1
         GO TO 118
  116    CONTINUE
  117 CONTINUE
  118 ID501=0
      ID502=0
      DO 121 I=1,5
         IF ((D50.GE.D50G(I)).AND.(D50.LE.D50G(I+1))) GO TO 119
         GO TO 120
  119    ID501=I
         ID502=I+1
         GO TO 122
  120    CONTINUE
  121 CONTINUE
  122 CONTINUE
      II(1)=ID1
      II(2)=ID2
      JJ(1)=IV1
      JJ(2)=IV2
      KK(1)=ID501
      KK(2)=ID502
      DO 130 I=1,2
         I1=II(I)
         XX(I)=ALOG10(DG(I1))
         DO 129 J=1,2
            J1=JJ(J)
            YY(J)=ALOG10(VG(J1))
         DO 129 K=1,2
            K1=KK(K)
            ZZ(K)=ALOG10(D50G(K1))
            IF (G(I1,J1,K1)-0.) 123,123,127
  123       DO 125 J3=J1,7
               IF (G(I1,J3,K1)-0.) 124,124,126
  124          CONTINUE
  125       CONTINUE
  126       X(J,K)=ALOG10(G(I1,J3,K1))+(ALOG10(VG(J1)/VG(J3)))*(ALOG1
     1      0(G(I1,J3+1,K1)/G(I1,J3,K1)))/(ALOG10(VG(J3+1)/VG(J3)))
            GO TO 128
  127       CONTINUE
            X(J,K)=ALOG10(G(I1,J1,K1))
  128       CONTINUE
  129    CONTINUE
         XD=ALOG10(D50)-ZZ(1)
         XN1=X(1,2)-X(1,1)
         XN2=X(2,2)-X(2,1)
         XDEN=ZZ(2)-ZZ(1)
         XA(1)=X(1,1)+XN1*XD/XDEN
         XA(2)=X(2,1)+XN2*XD/XDEN
         XNM=XA(2)-XA(1)
         XV=ALOG10(V)-YY(1)
         XDY=YY(2)-YY(1)
         XG(I)=XA(1)+XNM*XV/XDY
  130 CONTINUE
      XNM=XG(2)-XG(1)
      XD=ALOG10(D)-XX(1)
      XDEN=XX(2)-XX(1)
```

```
      GTUC=XG(1)+XNM*XD/XDEN
      GTUC=10.**GTUC
C
C     GTUC  IS UNCORRECTED GT IN LB/SEC/FT
C
C
C     NEXT APPLY  F.M.LOAD AND  TEMPERATURE CORRECTIONS
C
      IF (TF-60.) 132,131,132
  131 CFT=1.
      GO TO 137
  132 CONTINUE
      IT1=0
      IT2=0
      DO 135 I=1,6
         IF ((TF.GE.TEMP(I)).AND.(TF.LE.TEMP(I+1))) GO TO 133
         GO TO 134
  133    IT1=I
         IT2=I+1
         GO TO 136
  134    CONTINUE
  135 CONTINUE
  136 CONTINUE
      XT(1,1)=ALOG10(T(IT1,ID1))
      XT(2,1)=ALOG10(T(IT2,ID1))
      XT(1,2)=ALOG10(T(IT1,ID2))
      XT(2,2)=ALOG10(T(IT2,ID2))
      XNT=ALOG10(TF/TEMP(IT1))/ALOG10(TEMP(IT2)/TEMP(IT1))
      XCT(1)=XT(1,1)+XNT*(XT(2,1)-XT(1,1))
      XCT(2)=XT(1,2)+XNT*(XT(2,2)-XT(1,2))
      CFT=XCT(1)+(XCT(2)-XCT(1))*XD/XDEN
      CFT=10.**CFT
C
C     FINE MATERIAL LOAD  CORRECTION
C
  137 CONTINUE
      IF (FML-10.) 138,138,139
  138 CFF=1.
      GO TO 149
  139 CONTINUE
      IF (FML.GT.1.E+5) REMARK=5HFOOR
      ID1=0
      ID2=0
      DO 141 I=1,9
         IF ((D.GE.DF(I)).AND.(D.LE.DF(I+1))) GO TO 140
         GO TO 141
  140    ID1=I
         ID2=I+1
         GO TO 142
  141 CONTINUE
  142 CONTINUE
      IF (REMARK.EQ.5HFOOR    )143,144
  143 IF1=4
      IF2=5
      GO TO 148
  144 CONTINUE
      IF1=0
      IF2=0
      DO 147 I=1,4
         IF ((FML.GE.CF(I)).AND.(FML.LE.CF(I+1))) GO TO 145
         GO TO 146
  145    IF1=I
         IF2=I+1
         GO TO 148
  146    CONTINUE
  147 CONTINUE
  148 CONTINUE
      XF(1,1)=ALOG10(F(IF1,ID1))
      XF(2,2)=ALOG10(F(IF2,ID2))
      XF(1,2)=ALOG10(F(IF1,ID2))
      XF(2,1)=ALOG10(F(ID2,ID1))
      XNT=(FML-CF(IF1))/(CF(IF2)-CF(IF1))
      XCT(1)=XF(1,1)+XNT*(XF(2,1)-XF(1,1))
      XCT(2)=XF(1,2)+XNT*(XF(2,2)-XF(1,2))
      XNT=ALOG10(D/DF(ID1))/ALOG10(DF(ID2)/DF(ID1))
      CFF=XCT(1)+XNT*(XCT(2)-XCT(1))
      CFF=10.**CFF
  149 CONTINUE
      TCF=CFT*CFF-1.
      CFD=1.
      IF ((D50.GE.0.20).AND.(D50.LE.0.30)) GO TO 154
      IP1=0
      IP2=0
      DO 152 I=1,10
         IF ((D50.GE.DP(I)).AND.(D50.LE.DP(I+1))) GO TO 150
         GO TO 151
  150    IP1=I
         IP2=I+1
         GO TO 153
  151    CONTINUE
  152 CONTINUE
  153 CONTINUE
      P2=ALOG10(P(IP2))
      P1=ALOG10(P(IP1))
      XNT=ALOG10(D50/DP(IP1))/ALOG10(DP(IP2)/DP(IP1))
      CFD=P1+XNT*(P2-P1)
```

```
      CFD=10.**CFD
154   CONTINUE
      FFF=CFD*TCF
      FFF=FFF+1.
      GT=FFF*GTUC
      GT=GT*W
      WRITE (6,161) GT,REMARK
      GO TO 156
155   CONTINUE
      WRITE (6,158) REMARK
156   CONTINUE
157 CONTINUE
C
158 FORMAT (5X, 38HCOMPUTATIONS COULD NOT BE CARRIED OUT ,/5X, 24HDUE
   1TO DATA OUT OF RANGF,/5X,  8HREMARK= ,R10///)
159 FORMAT (1H1,9X, 33HCOMPUTATION OF TOTAL BED MATERIAL,/10X, 26HTRAN
   1SPORT BY COLBYS METHOD,//)
160 FORMAT (5X,4HSET ,I5/5X,27HAVERAGE VELOCITY            ,F12.2,12H F
   1T./SEC.    ,/5X,27HHYDRAULIC DEPTH             ,F12.2,12H FT.
   2 ,/5X,27HWATER SURFACE WIDTH         ,F12.2,12H FT.         ,/5X,27H
   3TEMPERATURE                ,F12.2,12H DEG.FAHREN.,/5X,27HMEDIAN BE
   4D MATERIAL SIZE    ,F12.2,12H MM.         ,/5X,27HFINE MATERIAL CONC
   5ENTRATION,F12.2,12H PPM.        ,/)
161 FORMAT (5X,24HBED MATERIAL TRANSPORT =,F15.5,12H TONS/DAY    ,/5X,9
   1HREMARK = ,R10///)
162 FORMAT (I5)
163 FORMAT (6F10.0)
C
    END
```